

Muster und Häufigkeit von Sensibilisierungen im Epikutantest: Eine SAS-Anwendung

Thomas Bruckner, Heinrich Dickel

Abteilung Klinische Sozialmedizin
Universitätsklinikum Heidelberg
Bergheimer Str. 58
69115 Heidelberg

thomas.bruckner@med.uni-heidelberg.de
heinrich.dickel@med.uni-heidelberg.de

Zusammenfassung

Allergische Spättyp-Reaktionen auf Umweltstoffe werden heute routinemäßig mit dem Epikutantest geprüft. Dabei kommt es nicht selten zu Reaktionen auf zwei oder mehrere Testsubstanzen. Die Frage nach der Häufigkeit und dem Muster solcher Mehrfachsensibilisierungen soll hier mit einem einfachen SAS-Makro beantwortet werden, welches im Wesentlichen nur auf Routinen des SAS-base Paketes zugreift.

Keywords: Epikutantest, Mehrfachsensibilisierungen, PROC FREQ, array.

1 Das Problem

Die Jahresprävalenz von allergischen Kontaktekzemen wird in der Bevölkerung auf 5-11% geschätzt [1]. Als Auslöser kommen verschiedenste Substanzen (Allergene) aus dem beruflichen wie auch außerberuflichem Umfeld in Betracht, mit denen ein direkter Haut- bzw. Schleimhautkontakt gegeben ist.

In der dermatologischen Routinediagnostik erfolgt der Nachweis einer Typ IV-Sensibilisierung (Spättyp-Reaktion) gegen ein Allergen nach wie vor am zuverlässigsten durch den Epikutantest (Läppchentest, „patch test“) [2]. Dabei werden mehrere, nach Anamnese ausgewählte Testsubstanzen gleichzeitig auf

die Rückenhaut des Patienten appliziert, wobei oftmals Hautreaktionen, d.h. Sensibilisierungen auf eine oder mehrere der Testsubstanzen beobachtet werden können. Die Testreaktionen werden je nach Schweregrad mit 0, ?/+, +, ++, +++ oder IR (irritative, d.h. keine allergische Reaktion) bewertet.

Findet man mehrere allergische Reaktionen (Mehrfachsensibilisierungen) bei Patienten, so stellt sich die Frage, ob sich gewisse Muster bei diesen Mehrfachsensibilisierungen zeigen oder ob die Verteilung als zufällig zu betrachten ist. Erschwerend kommt hinzu, dass eine Mehrfachsensibilisierung relativ selten ist und zur Zeit mehrere hundert Allergene als standardisierte Testsubstanzen eingesetzt werden.

Die Daten von 2881 epikutan getesteten Patienten [3] wurden reanalysiert. Die Ergebnisse von 17 Allergenen des Epikutantests wurden dichotomisiert (0, ?/+, IR = 0; +, ++, +++ = 1) und in 17 Variablen x_i gespeichert. Die mögliche Anzahl von Kombinationen beträgt bei k dichotomen Variablen 2^k , eine Zahl, die mit wachsendem k schnell groß wird. In unserem Beispiel mit 17 Allergenen beträgt die Anzahl der möglichen Kombinationen $2^{17} = 131072$.

2 Drei Lösungsansätze

Ein erster Versuch, die Frage nach „check all that apply“ zu beantworten, war die Lösung mit PROC FREQ und der LIST-Option im TABLE Statement [4].

```
proc freq ;  
    table x1*x2*...*x17 / list ;  
run ;
```

Diese Lösung ist für kleine Anzahlen von dichotomen Variablen geeignet, sie ist allerdings von der Ausstattung des Rechners abhängig. Unsere Konfiguration mit 384 MB Hauptspeicher gab bei 30 Variablen mit der Fehlermeldung „ERROR: The requested table is too large to process“ auf.

Eine zweite Möglichkeit ist die Bearbeitung des Problems mit einem Makro von Ronald Fehd (%checkall) [5]. Auch damit bekommt man für bis zu 50 Variablen eine Lösung, $k = 51$ ist bei diesem Makro allerdings (konstruktionsbedingt) die maximale Variablenzahl.

Wir haben uns deshalb entschlossen, eine eigene Lösung zu erstellen. Die Lösung sollte dabei recht einfach sein und möglichst nur mit Base-SAS [6] auskommen. Die Idee ist, die k dichotomen Variablen durch einen String abzubilden und dann als eindimensionale Tabelle auszuwerten.

Unsere Umsetzung in SAS erfolgt in 2 Schritten:

Pro Beobachtung wird die Anzahl der Reaktionen, $n_{reakt} = \sum x_i$ bestimmt,

der Inhalt der x_i in einen String der Länge k (im Beispiel ist $k = 17$) geschrieben und die Variablenamen der positiven Reaktionen, getrennt durch Komma, in eine Variable *combinations* geschrieben.

Im 2. Schritt erfolgt die Auswertung der aggregierten Daten:

Mit PROC FREQ wird

- (a) die Häufigkeit der multiplen Reaktionen berechnet (table *nreakt*), und
- (b) die Häufigkeit der einzelnen Kombinationen (table *combinations*) bestimmt.

Man kann die gefundene Anzahl der Kombinationen nun graphisch darstellen, die Häufigkeit der positiven Einzelreaktionen der Allergene aufzeigen und die gefundenen Allergenkombinationen (sortiert) auflisten.




Abbildung 1: Häufigkeit von Patienten ohne, mit einer und mit mehreren Sensibilisierungen

3 Diskussion

Es wurde eine einfache Lösung des Problems „check all that apply“ [5] vorgestellt. Das Makro ist noch in vielen Richtungen erweiterungsfähig. So kann man die graphische Ausgabe integrieren, fehlende Werte in den Variablen berücksichtigen und an eine Erweiterung auf Variablen mit mehr als 2 Ausprägungen denken.

Der Grundgedanke allerdings, dass man mit SAS komplexere Fragestellungen durchaus nur mit SAS-base Routinen lösen kann, ist klar gezeigt worden.

Literatur

- [1] Schnuch, A. (1991) Zur Epidemiologie des Kontaktekzems. In: Ring, J. (eds) Epidemiologie allergischer Erkrankungen: Nehmen Allergien zu? MMV Medizin Verlag-Vieweg, München, pp. 37-82
- [2] Fuchs, T., Gutgesell, C. (2000) Epikutantest. In: Przybilla, B., Bergmann, KC., Ring, J. (eds.) Praktische allergologische Diagnostik. Steinkopff Verlag, Darmstadt, pp. 23-39
- [3] Dickel, H., Taylor, J. S., Bickers, D. R., Rehse, E., Merk, H. F. Patch testing with a standard series: results from two clinics in Cleveland, Ohio (USA), and comparison with a West European clinic in Cologne, North Rhine-Westphalia (Germany). *Dermatosen* 1998; 46(6): 234-243
- [4] SAS Institute Inc. The FREQ procedure. In: SAS Institute Inc., ed. SAS/STAT User's Guide, Version 8. 1st ed. Cary, North Carolina: SAS Institute Inc., 1999: 1245-1361
- [5] Fehd, R. %SHOWCOMB: a macro to produce a data set with frequency of combinations of responses from multiple-response data. Proceedings of the Twenty-Second Annual SAS Users Group International Conference. San Diego, California: SAS Institute, Inc. (SAScontrib@SASserv.uga.edu), 1997: 939-943
- [6] SAS Institute Inc., SAS Language Reference: Concepts. Cary, North Carolina: SAS Institute Inc., 1999. 554 pages

Anhang

SAS-Code

```

/**      Makro zur Bestimmung aller möglichen Kombinationen
        von k dichotomen Variablen
**/
%macro checkit(_data=, /*Input data set*/
              _vars=, /* List of variables*/
              _nn=,   /*Number of variables */
              _text1=); /*Text to appear in Output file*/
data test ;
  set &_data ;
  length
    check $ &_nn
    testit $ 50
    t $8 ;
  array xx(&_nn) &_vars;
  ** berechne Anzahl der Reaktionen pro Beobachtung ;
  nreakt=0 ;
  do i = 1 to &_nn ;
    nreakt=nreakt+xx(i) ;
  end;
  ** erzeuge String ;
  do i=1 to &_nn ;
    check = compress(check) || put(xx(i),1.) ;
    if substr(check,i,1) = '1' then do;
      call vname(xx(i),t);
      testit=compress(testit)|| t ||', ';
    end;
  end;
end;

testit=compress(testit);
** Hilfsvariable mit den Kombinationen drin ;

** check all that apply ;
** in combis stehen die Kombinationen ;
** bestimme Kombinationen ;
if length(testit) > 0 then
  combis=substr(testit,1,length(testit)-1) ;
  ** letztes komma ueberfluessig ;
if combis=' ' then combis = '0' ; ** statt blank ;
kk=input(&_nn,best.);
two_k = 2**kk ;
run;

```

```
proc freq noprint data=test ;
    tables nreakt / out=nreact ;
run;

proc means noprint data=test ;
    var &_vars;
    output out = _crude sum = &_vars ;
run;

data _crude (keep=_name_ merge value perc );
    set _crude ;
    length param $40 _name_ $8 ;
    label param = 'Allergen'
           value = 'pos. Reactions'
           perc = 'Percent' ;
    array xx &_vars ;
    array pp(&_nn) p1 - p&_nn ;
    do merge = 1 to &_nn ;
        pp(merge) = round(xx(merge) /_freq_ * 100,0.01) ;
        value = xx(merge);
        perc = pp(merge) ;
        call vname(xx(merge),_name_);
        output ;
    end;
run;

data nobs;
    set test end=eof ;
    nobs=_n_ ;
    if eof then output ;
run;

data label (keep=&_vars) ;
    set test ;
    if _n_ = 1;
run;

proc transpose data=label out = _label ;
data _label ;
    set _label ;
    merge =_n_ ;
run;

data _label ;
    merge _label _crude ;
    by merge ;
run;
```

```

proc freq data=test ;
    table combis / missing noprint out = result;
run;

data result ;
    set result ;
    dummy=length(combis);
run;

proc sort data=result ;
    by dummy combis;
run;

/*****
/**
    output section
    Realisation als .txt Datei
    filename results ausserhalb des Makros definiert
**/

/**
    allgemeine Informationen
**/
data _null_ ;
    file results ;
    set nobs ;
    put "&_text1" ;
    put ' ' ;
    put 'Datensatzbeschreibung' ;
    put ' ' ;
    put nobs:8. 'Beobachtungen, ' "&_nn Allergene, " ;
    put two_k" mögliche Kombinationen";
run;

/*
    Variablenbeschreibung - Kopf
*/
data _null_ ;
    file results mod ;
    put ' ' ;
    put 'Variablenbeschreibung' ;
    put ' ' ;
    put 'Variable' @20 'Beschreibung' @60 'N positiv' @70 '%' ;
    put ' ' ;
run;

```

```
/*
  Variablenbeschreibung - Tabelle
*/
data _null_ ;
  set _label ;
  File results mod ;
  put @5 _name_ @20 _label_ @60 value @70 perc:6.2;
run;

/*
  Variablenbeschreibung - Anzahl Kombinationen
*/
data _null_ ;
  set nreact ;
  file results mod ;
  if _N_ =1 then do ;
    put ' ';
    put 'Zusammenfassung der Ergebnisse' ;
    put ' ';
    put 'Anzahl Kombinationen' @30 'N' @40 '%' ;
    put '-----';
  end;
  put nreakt @30 count @40 percent:6.2 ;
run;

/**
  Allergenkombinationen Tabellenkopf
**/
data _null_ ;
  file results mod;
  put ' ';
  put 'Vorkommen und Anzahl Kombinationen' ;
  put ' ';
  put 'Kombinationen                Anzahl    Prozent' ;
  put '-----';
run;

/**
  Allergenkombinationen Ergebnisse
**/
data _null_ ;
  set result ;
  File results mod ;
  put combis @60 count @70 percent 6.2;
run;
%mend checkit ;
```

```

/* Hauptprogramm*/
Libname in "J:\ksfe\checkall" ;

filename results "J:\ksfe\checkall\result.txt" ;

%include "J:\ksfe\checkall\macro___checkit.sas" ;

options mprint symbolgen ;
options nocenter ;

Data a (keep=patno x1-x17) ;
    set in.all ;
run;

%checkit(_data=a,_vars=x1-x17,_nn=17,_text1=Allergen Test (Daten Dr. Dickel));

```

SAS-Output

Allergen Test (Daten Dr. Dickel)

Datensatzbeschreibung

2881 Beobachtungen, 17 Allergene,
131072 mögliche Kombinationen

Variablenbeschreibung

Variable	Beschreibung	N positiv	%
x1	Nickel sulphate	372	12.91
x2	Balsam of peru	169	5.87
x3	Formaldehyde	162	5.62
x4	Potassium dichromate	128	4.44
x5	Neomycin sulphate	121	4.20
x6	Paraphenylenediamine	112	3.89
x7	Quaternium-15	85	2.95
x8	Benzocaine	75	2.60
x9	Thiuram mix	73	2.53
x10	Colophony	64	2.22
x11	Carba mix	48	1.67
x12	Wool alcohol	45	1.56
x13	PPD mix	36	1.25
x14	Expoy resin	31	1.08
x15	Ethylendiamine	30	1.04
x16	Mercapto mix	22	0.76
x17	p-tert Butylphenol	14	0.49

Zusammenfassung der Ergebnisse

Anzahl Kombinationen	N	%
0	1819	63.14
1	706	24.51
2	247	8.57
3	69	2.40
4	25	0.87
5	11	0.38
6	3	0.10
7	1	0.03

Vorkommen und Anzahl Kombinationen

Kombinationen	Anzahl	Prozent
0	1819	63.14
X1	238	8.26
X2	79	2.74
X3	66	2.29
X4	45	1.56
X5	60	2.08
X6	41	1.42
X7	25	0.87
X8	25	0.87
X9	24	0.83
X10	27	0.94
X11	11	0.38
X12	15	0.52
X13	11	0.38
X14	13	0.45
X15	15	0.52
X16	4	0.14
X17	7	0.24
X1, X2	1	0.35
X1, X3	13	0.45
X1, X4	23	0.80
....		
....		
X2, X4, X6, X7, X9, X10	1	0.03
X2, X5, X8, X9, X12, X16	1	0.03
X4, X6, X9, X11, X13, X16	1	0.03
X1, X2, X4, X6, X7, X9, X15	1	0.03